﻿* Encoding: UTF-8.

/* Jones_Adviser to the King: Experts, Rationalization and Legitimacy

/*Study 2 Nationationality of Experts  
/*Datafile Jones_Study2.sav

VALUE LABELS
Condition
1 'American'
2 'Kuwaiti'
3 'Chinese'.
Execute.

VALUE LABELS
religiosity
1 ' Religious'
2 'Somewhat religious'
3 'Not religious'.
Execute.

VALUE LABELS
parented
mothered
fathered
1 'None'
2 'Elementary or primary'
3 'Secondary'
4 'BA or college diploma'
5 'MA or higher'.
Execute.

VALUE LABELS
gender
1 'Female'
2 'Male'.
Execute.

VALUE LABELS
income
1 'Low'
4 'Middle'
7 'High'. 
Execute.

VARIABLE LABELS
supportplan 'Do you support or oppose the plan proposed by politicians [and international experts]'
pctsupportplan 'What percentage of Kuwait’s population do you think will support the plan proposed by politicians [and international experts]?'
pctsucceed 'On average, what percentage of projects like this would you say succeed?'
succeedcertain 'How certain are you that the plan proposed by the political leaders [and international experts] will succeed?'
lovecountry 'I love my country'
proudcitizen 'I am proud to be a citizen of my country'
techoptimism 'How confident are you that advanced technologies will improve global healthcare within the next five years?'
achoptimism 'If you had to guess, how many Nobel Prize winners do you think Kuwait will produce over the next 10 years? Enter a number.'
parented 'Parental education'.
Execute.

compute parented = mean(fathered,mothered).

compute supportplan_01 = supportplan/7.
compute pctsupportplan_01 = pctsupportplan/11.
compute pctsucceed_01 = pctsucceed/11.
compute succeedcertain_01 = succeedcertain/7.
compute lovecountry_01 = lovecountry/7.
compute proudcitizen_01 = proudcitizen/7.

compute support_index_01 = mean(supportplan_01,pctsupportplan_01).
compute success_index_01 = mean(pctsucceed_01,succeedcertain_01).
compute patriot_index_01 = mean(lovecountry_01,proudcitizen_01).
compute techoptimism_01 = techoptimism/7. 
Execute.


VARIABLE LABELS
support_index_01 'Support for reform'
success_index_01 'Confidence in reform'
patriot_index_01 'Patriotism'
techoptimism_01 'Optimism about technological progress'
achoptimism 'Optimism about human achievement (Kuwaiti Nobel Prize winners)'.
Execute.


/*Demographic balance tests

ONEWAY age religiosity income parented BY Condition
  /STATISTICS DESCRIPTIVES
  /MISSING ANALYSIS.
Execute.

/*One-way ANOVA on dependent variables and pairwise post hoc comparisons

ONEWAY support_index_01 success_index_01 patriot_index_01 techoptimism_01 achoptimism BY Condition
  /STATISTICS DESCRIPTIVES HOMOGENEITY
  /MISSING ANALYSIS
  /POSTHOC=TUKEY ALPHA(0.05).

/*T-test for paired samples to compare subjects' overall support for the plan  to their estimated confidence in its success. 

T-TEST PAIRS=support_Index_01 WITH success_index_01 (PAIRED)
  /CRITERIA=CI(.9500)
  /MISSING=ANALYSIS.


/* Generate SPSS graph (Figure 2) (Note: Final graph appearing in paper was created in Excel)

* Chart Builder.
GGRAPH
  /GRAPHDATASET NAME="graphdataset" VARIABLES=Condition
    MEAN(support_index_01)[name="MEAN_support_index_01"] MISSING=LISTWISE REPORTMISSING=NO
  /GRAPHSPEC SOURCE=INLINE.
BEGIN GPL
  SOURCE: s=userSource(id("graphdataset"))
  DATA: Condition=col(source(s), name("Condition"), unit.category())
  DATA: MEAN_support_index_01=col(source(s), name("MEAN_support_index_01"))
  GUIDE: axis(dim(1), label("Condition"))
  GUIDE: axis(dim(2), delta(.05), label("Mean Support for Reform"))
  SCALE: cat(dim(1), include("1", "2", "3"))
  SCALE: linear(dim(2), min(.40), max(.75))
  ELEMENT: interval(position(Condition*MEAN_support_index_01), shape.interior(shape.square))
END GPL.


/* Generate SPSS graph (Figure 3) (Note: Final graph appearing in paper was created in Excel)

* Chart Builder.
GGRAPH
  /GRAPHDATASET NAME="graphdataset" VARIABLES=Condition
    MEAN(success_index_01)[name="MEAN_success_index_01"] MISSING=LISTWISE REPORTMISSING=NO
  /GRAPHSPEC SOURCE=INLINE.
BEGIN GPL
  SOURCE: s=userSource(id("graphdataset"))
  DATA: Condition=col(source(s), name("Condition"), unit.category())
  DATA: MEAN_success_index_01=col(source(s), name("MEAN_success_index_01"))
  GUIDE: axis(dim(1), label("Condition"))
  GUIDE: axis(dim(2), label("Mean Confidence in Reform"))
  SCALE: cat(dim(1), include("1", "2", "3"))
  SCALE: linear(dim(2), min(.40), max(.75))
  ELEMENT: interval(position(Condition*MEAN_success_index_01), shape.interior(shape.square))
END GPL.

